
ABSTRACT  
GoodCents contracted with Mississippi Power to develop the Rate Manager System, a user friendly application 
designed for non- SAS®  users that allow the user to design Utility Rates and to do a myriad of bill comparisons with 
other Utility rates. The software has a number of menu-driven tasks as well as 12 graphics applications to allow visual 
comparisons of Rates. The system was developed primarily using Proc Pmenu but also uses SAS®  Graph, FSVIEW, 
Proc Surveyselect as well as extensive use of compiled macros to compute customer bills for a variety of Utility rate 
types from simple to complex. The software was presented at the Southeastern Electric Exchange's Rate section in 
2005 and won first place. We will present the software from a users design perspective and from a coders perspective 
and share some of the code out of the approximately 100,000 lines of code and over 200 SAS®  programs which are 
linked together to form the Rate Manager System. 

INTRODUCTION  
The Rate Manager system was developed to assist the Rate Analyst in the analysis of current rates and the design of 
new rates. It also allows for rate comparisons across companies and jurisdictions. The current application is a rewrite 
in SAS® (Statistical Analysis System) of the previous DOS-based system. The rewrite was undertaken to update the 
system to a Microsoft Windows compatible point-and-click environment with pull-down menus, easy exchange 
between applications, and ease of use for printing copy and graphics. Mississippi Power Company (MPC) retained 
Energy Sciences Group (d/b/a GoodCents Solutions) in September 2002 to redesign and rewrite the DOS-based 
system using SAS®. The GoodCents project manager and programmer is Bill Bland and the MPC project manager is 
Allen Dunn. The first part of this paper will present part of the menu structure and code detail. The second part of the 
paper will present an overview of the system and some results. 

 MENU STRUCTURE AND SAMPLE MENUS 
The development of the menu structure developed from discussions with the users and from the old DOS based 
system. We used Proc Pmenu to develop the following drop down menus:  
Rates/Adder
s 

FORMUL
A 

Differential 
Chart 

Graph   Bill 
Comparison 

Hel
p 

The Rates/Adders drop down box allows the user to input new rates using a template, modify existing rates or adders.  
This menu is shown below. 

The Formulization Menu shown below allows the user to select a rate along with various bill adders for Formulization 
by selecting the appropriate radio buttons. Once the user selects OK the Formula SAS® code is invoked. 
The FORMULA routine develops short-form rate equations for an input rate. For a given KW-KWH range, a rate’s 
Price of Energy (Bill) can be represented as a linear equation of the form: 
 
Bill = A*Energy + B*Demand + Constant 
The A & B terms and Constants are developed by FORMULA for the appropriate KW-KWH range.  
The FORMULA routine will also automatically apply the selected rate adders to the rate equations.  

The Differential Chart Menu shown below invokes a SAS® program that calculates the line or lines of equality 
between the 2 selected rates (if one exists).  The user selects one rate from group 1 and one rate from group 2 by 
selecting (pushing) the radio button and then selecting OK. The Graph Menu, Bill Comparison, Rate Design and Help 
Menus are presented to the user in a similar manner . 

SAS®  CODE FOR MENU DEVELOPMENT 
 
The system is initially invoked by running the driver program that displays an introductory page and invokes the main 
menu. The driver program is shown below. 

 
/******************************************************************************/ 
/*  DATE CREATED:  September 2002                                             */ 
/*  BY:  Bland                                                                */ 
/******************************************************************************/ 
/*  DESCRIPTION:                                                              */ 
/*    THIS PROGRAM is the initial driver program for the Rateman system.      */ 
/*    It calls the main menu and the initial display screen                   */ 
/******************************************************************************/ 
/*  INPUTS:                                                                   */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  OUTPUTS:                                                                  */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  EDITED BY:                             DATE:                              */ 
/******************************************************************************/ 
options nomlogic nomprint nomrecall nosymbolgen; 
/**** mpcprod is the main directory that the Rateman system is installed in ***/ 
%global maindir; 
%let maindir=c:\mpcprod; 
proc printto log="&maindir\log_file.txt" new; 
run; 
/********rateman-pmenu.sas is the initial menu for the system*******************/ 
filename menu "&maindir\rateman-pmenu.sas"; 
%include menu; 
 
 
run; 
 
/************mainwin-rateman.sas is the file storing the initial screen verbage**/ 
filename hrwdw "&maindir\mainwin-rateman.sas"; 
   %include hrwdw; 
   run; 

The introductory page is displayed using the mainwin-rateman program shown below. 
 
/******************************************************************************/ 
/*  PROGRAM:  mainwin-rateman                                                 */ 
/******************************************************************************/ 
/*  DATE CREATED:  September 2002                                             */ 
/*  BY:  Bland                                                                */ 
/******************************************************************************/ 
/*  DESCRIPTION:                                                              */ 
/*    THIS PROGRAM stores the iformation shown on the initial display screen  */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  INPUTS:                                                                   */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  OUTPUTS:                                                                  */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  EDITED BY:                             DATE:                              */ 
/******************************************************************************/ 
  data _null_; 
      window RATEMAN menu=proclib.rateman.rateman 
   #2  @10 'MISSISSIPPI POWER COMPANY RATE MANAGER SYSTEM - V2006.01' 
      #4  @10 'This application develops short form rate equations' 
      #5  @10 'for the Rate selected and lines of equality between 2 selected rates' 
      #6  @10 'It also displays a graphic comparison for 2 rates selected' 
      #7  @10 '                   ' 
      #9  @10 'Select Rates/ Adders Menu on command line' 
      #10 @10 'for Rate Edits or to End the application' 
      #11 @10 'or Exit from the SAS System.' 
   #12 @10 '                   ' 
   #13 @10 'Select Formula on command line' 

#14 @10 'to select rates, adders or power factor for development of short form 
rate equations or' 

      #15 @10 'to End the application or Exit from the SAS System.' 
   #16 @10 '                   ' 
   #17 @10 'Select Differential Chart on command line' 

#18 @10 'to select rates for development of lines of rate equality between the 
2 rates selected or' 

      #19 @10 'to End the application or Exit from the SAS System.' 
   #20 @10 '                   ' 
   #21 @10 'Select Graph on command line' 

#22 @10 'to select rates for graph of lines of rate equality between the 2 
rates selected or' 

      #23 @10 'to End the application or Exit from the SAS System.' 
   #24 @10 '                   ' 
   #25 @10 'Select Bill Comparison on command line' 

#26 @10 'to select rates for bill comparisons at various demand and energy 
levels or' 

      #27 @10 'to End the application or Exit from the SAS System.' 
   #28 @10 '                   ' 
   #29 @10 '                   ' 
   #30 @10 'Select Rate Design on command line' 
      #31 @10 'to select rates for billing determinants entry and rate design or' 
      #32 @10 'to End the application or Exit from the SAS System.' 
   #33 @10 '                   ' 
   #34 @10 '                   ' 

#35 @10 ' !!!!! ON INITIAL SYSTEM STARTUP SELECT: INITIALIZE\ UPDATE FILE MENU 
!!!!! ' ; 

       display RATEMAN; 
   run; 

The SAS code for development of the main menu system displayed above is shown below. 
 

/******************************************************************************/ 
/*  PROGRAM:  rateman-pmenu                                                   */ 
/******************************************************************************/ 
/*  DATE CREATED:  September 2002                                             */ 
/*  BY:  Bland                                                                */ 
/******************************************************************************/ 
/*  DESCRIPTION:                                                              */ 
/*    THIS PROGRAM is the initial menu program for the Rateman system.        */ 
/*    It sets up the main pull down menu                                      */ 
/******************************************************************************/ 
/*  INPUTS:                                                                   */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  OUTPUTS:                                                                  */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  EDITED BY:                             DATE:                              */ 
/******************************************************************************/ 
options nomlogic nomprint nomrecall nosymbolgen; 
/***the main procedure library is stored in the &maindir directory*************/ 
/****the &maindir directory is the main directory for the system specified in the 
initial driver program*****/ 
libname proclib "&maindir"; 
 
/*********the fileinit statement refers to the main setup file for directories, 
libraries, etc.**************/ 
filename fileinit "&maindir\rateman-file-setup.sas";  
filename finitmpc "&maindir\rateman-file-setup-mpc.sas";  
 
/***Proc PMENU is the SAS procedure that allows menus to be set 
up***********************/ 
proc pmenu catalog=proclib.rateman; 
 
/******the rateman menu controls the main 
dropdowns***********************************/ 
 
     menu rateman; 
/*****Rates/Adders is the first dropdown defined***************/ 
        item 'Rates/ Adders' menu=f; 
  item 'Formula' menu=g; 
  item 'Differential Chart' menu=h; 
  item 'Graph' menu=j; 
  item 'Bill Comparison' menu=l; 
  item 'Rate Design' menu=m; 
  item 'Help' menu=k; 
/***** menu f allows for transfer to the rateadder submenu, file initialization, end 
program, end SAS or clear Log file*/ 
        menu f; 
  item 'Select Rates/ Adders' submenu=rateadder; 
  item 'Initialize\Update File Menu' selection=fileupdt;  
  item 'Initialize\Update File Menu-MPC Only' selection=fupdtmpc;  
        item 'End this window' selection=endwdw; 
        item 'End this SAS session' selection=endsas; 
     item 'Clear Log' selection=clrlog;  
  selection fileupdt 'end; pgm; include fileinit;submit'; 
  selection fupdtmpc 'end; pgm; include finitmpc;submit'; 
        selection endwdw 'end'; 
        selection endsas 'bye'; 

     selection clrlog 'clear log'; 
 
 
          submenu rateadder proclib.rateman.rateadder; 
/*menu g allows for transfer to companyf menu for rate formulization*/ 
     menu g; 
  item 'Select Rates/Adders for Formulization' submenu=companyf;  
  item 'End this window' selection=endwdw; 
        item 'End this SAS session' selection=endsas; 
  selection endwdw 'end'; 
        selection endsas 'bye'; 
   
  submenu companyf proclib.rateman.companyf;  
 
    
  menu h;  
  item 'Select Rates for Diffchart' submenu=diff;  
  item 'End this window' selection=endwdw; 
        item 'End this SAS session' selection=endsas; 
  selection endwdw 'end'; 
        selection endsas 'bye'; 
  submenu diff proclib.rateman.companydiff;  
 
  menu j; 
  item 'Select Diffchart File for Graph' submenu=graph;   
  item 'End this window' selection=endwdw; 
        item 'End this SAS session' selection=endsas; 
  selection endwdw 'end'; 
        selection endsas 'bye'; 
     submenu graph proclib.rateman.companygph;  
 
  menu k;  
  item  'Select Help Topic' submenu=help;  
  item 'End this window' selection=endwdw; 
        item 'End this SAS session' selection=endsas; 
  selection endwdw 'end'; 
        selection endsas 'bye'; 
  submenu help proclib.rateman.help;  
 
  menu l;  
  item 'Select Rates for Bill Comparison' submenu=rcomp;  
  item 'End this window' selection=endwdw; 
        item 'End this SAS session' selection=endsas; 
  selection endwdw 'end'; 
        selection endsas 'bye'; 
  submenu rcomp proclib.rateman.companycomp;  
 
  menu m;  
  item 'Select Rates for Rate Design' submenu=rdes;  
  item 'End this window' selection=endwdw; 
        item 'End this SAS session' selection=endsas; 
  selection endwdw 'end'; 
        selection endsas 'bye'; 
  submenu rdes proclib.rateman.rdesign; 
quit; 

 
On first running the system the user selects 'Initialize\Update File Menu-MPC Only'.  This transfers control to SAS  
code, rateman-file setup-mpc that initializes libnames and input directories where Rate files and other files are stored. 
We create macro variables for use as  libnames in other programs. This way for instance we can change the main 
production directory from c:\mpcprod to h:\mpcprod in this one program and the change is reflected in the hundreds of 
other programs used by the system. This SAS®  code is shown below. 

/******************************************************************************/ 
/*  PROGRAM:  rateman-file setup-mpc                                          */ 
/******************************************************************************/ 
/*  DATE CREATED:  October 2002                                               */ 
/*  BY:  Bland                                                                */ 
/******************************************************************************/ 
/*  DESCRIPTION:                                                              */ 
/*    THIS PROGRAM controls the Rateman system file setups for MPC only       */ 
/*    It contains file directories, libraries and program calls               */ 
/*    to update menus                                                         */ 
/******************************************************************************/ 
/*  INPUTS:                                                                   */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  OUTPUTS:                                                                  */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  EDITED BY:    Bland                         DATE:2/2006                   */ 
/******************************************************************************/ 
 
/***the main procedure library is stored in the &maindir directory*************/ 
/****the &maindir directory is the main directory for the system specified in the 
initial driver program*****/ 
%let maindir=c:\mpcprod; 
 
/* These are the DOS file directory statements stored in macro variables */ 
/* used in the pipe commands in the ratefilelist programs- 2 statements per company 
active and historical*/ 
%let filedirma=dir/b &maindir\rates\mpc\active\*rat.sas7bdat; 
%let filedirmao=dir/b &maindir\rates\mpc\active\mpc*out.sas7bdat; 
%let filedirmh=dir/b &maindir\rates\mpc\history\*rat.sas7bdat; 
%let filedirmg=dir/b &maindir\difffile\mpc*diff.sas7bdat; 
%let fildirmaaddrs=dir/b &maindir\rates\mpc\active\adders\*.sas7bdat; 
%let fildirmhaddrs=dir/b &maindir\rates\mpc\history\adders\*.sas7bdat; 
%let fildirgif=dir/b &maindir\weboutput\diffchart\*.gif; 
%let fildirbas=dir/b &maindir\customer-kw-kwh\*bas.csv; 
/* These are the file directories stored in macro variables */ 
/* 2 statements per company active and historical */  
%let vfiledirma=&maindir\rates\mpc\active\;  
%let vfiledirmh=&maindir\rates\mpc\history\;  
%let vfiledirmc=&maindir\customer-kw-kwh\;  
 
%let vfildirmaaddrs=&maindir\rates\mpc\active\adders;  
%let vfildirmhaddrs=&maindir\rates\mpc\history\adders;  
%let vfilediff=&maindir\difffile; 
%let vfileout=&maindir\outfile;  
 
/* These are the libname statements for the file directories */ 
/* 2 statements per company active and historical */  
libname mpcactiv "&vfiledirma"; 
libname mpchist "&vfiledirmh"; 
 
libname mactadd "&vfildirmaaddrs"; 
libname mhisadd "&vfildirmhaddrs"; 
libname diff "&vfilediff"; 
libname main "&maindir"; 
 
/*These macro variables contain the files directories for the adder and rate 
template files*/ 
%let adder=&maindir\adders\; 

%let ratetemp=&maindir\ratetemp\; 
 
/*These filename and include ststements run the ratefilelistXX programs to update 
the */  
/*macro variables used to define the file selection menus */ 
/* 2 statement groups per company active and historical */  
filename flistma "&maindir\ratefilelistma.sas"; 
%include flistma; 
filename flistmao "&maindir\ratefilelistmao.sas"; 
%include flistmao; 
filename flistmh "&maindir\ratefilelistmh.sas"; 
%include flistmh; 
filename fladdma "&maindir\ratefilelist-adders-ma.sas"; 
%include fladdma; 
filename fladdmh "&maindir\ratefilelist-adders-mh.sas"; 
%include fladdmh;  
filename flistmg "&maindir\ratefilelistmgph.sas"; 
%include flistmg; 
filename flistgif "&maindir\ratefilelistgif.sas"; 
%include flistgif; 
filename flistbas "&maindir\ratefilelistbas.sas"; 
%include flistbas; 
 
 
/*These filename and include ststements run the ratetypeXXXX programs to update the 
*/  
/* the file selection menus */ 
/* 2 statement groups per company active and historical */  
filename mpc "&maindir\rateman-pmenu-ratetypempc.sas"; 
 filename mpcf "&maindir\rateman-pmenu-ratetypempc-formula.sas"; 
filename mpch "&maindir\rateman-pmenu-ratetypempch.sas"; 
 filename mpchf "&maindir\rateman-pmenu-ratetypempch-formula.sas"; 
filename mpcd "&maindir\rateman-pmenu-ratetypempc-diffchart.sas"; 
filename mpcg "&maindir\rateman-pmenu-ratetypempc-graph.sas"; 
filename compa "&maindir\rateman-pmenu-ratetypeactive-ratecomp.sas"; 
filename xsec "&maindir\rateman-pmenu-ratetypeactive-graphxsection.sas"; 
filename compit "&maindir\rateman-pmenu-ratetypeactive-ratecomp-level.sas"; 
 
%include mpc; 
%include mpcf; 
%include mpch; 
%include mpchf; 
%include mpcd; 
%include mpcg; 
%include compa; 
%include xsec; 
%include compit; 
 
/*These filename and include statements run the rateadder, rateman and company menu 
programs to update */  
/* those menus */ 
 
filename rateadd "&maindir\rateman-pmenu-rateadder.sas"; 
filename main "&maindir\rateman-pmenu.sas"; 
filename company "&maindir\rateman-pmenu-company.sas"; 
filename companyf "&maindir\rateman-pmenu-company-formula.sas"; 
filename companyd "&maindir\rateman-pmenu-company-diff.sas"; 
filename companyg "&maindir\rateman-pmenu-company-graph.sas"; 
filename companyc "&maindir\rateman-pmenu-company-ratecomp.sas"; 
filename rmmhelp "&maindir\rateman-pmenu-help.sas"; 
filename rmrdes "&maindir\rateman-pmenu-ratedesign.sas"; 
%include rateadd; 
%include main; 
%include company; 
%include companyf; 

%include companyd; 
%include companyg; 
%include companyc; 
%include rmmhelp; 
%include rmrdes; 
run; 
 
/*The filename and include statements return control to the driver program*/ 
filename driver "&maindir\rateman-driver.sas"; 
%include driver; 

  

MPC RATE SELECTION MENU 
If the user selects the MPC active rate selection, the following menu code is used. The dialog statements call a 
program that  invokes PROC FSVIEW and allows the user to view and edit MPC rate files that are stored in Windows 
directories. The MPC Rate menu code is shown below. 
 

/******************************************************************************/ 
/*  PROGRAM:  rateman-pmenu-ratetypempc                                       */ 
/******************************************************************************/ 
/*  DATE CREATED:  September 2002                                             */ 
/*  BY:  Bland                                                                */ 
/******************************************************************************/ 
/*  DESCRIPTION:                                                              */ 
/*    THIS PROGRAM controls the MPC Active Rates selection submenu            */ 
/*    for the Rateman system.                                                 */ 
/*    It allows for selection of active MPC rates and editing of those rates. */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  INPUTS:                                                                   */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  OUTPUTS:                                                                  */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  EDITED BY:                             DATE:                              */ 
/******************************************************************************/ 
options nomlogic nomprint nomrecall nosymbolgen; 
/***the main procedure library is stored in the &maindir directory*************/ 
/****the &maindir directory is the main directory for the system specified in the 
initial driver program*****/ 
libname proclib "&maindir"; 
/**the fileviewma program allows the user to edit MPC active rate files*******/ 
filename fileview "&maindir\fileviewma.sas"; 
 
/**the ratefilelistma program updates the rate selection menu for rates found in the 
MPC active file directory*******/ 
filename filelist "&maindir\ratefilelistma.sas"; 
 
/***Proc PMENU is the SAS procedure that allows menus to be set 
up***********************/ 
proc pmenu catalog=proclib.rateman; 
 
/******the ratetypempc menu controls the selection of the MPC active rate file for 
editing*/ 
     menu ratetypempc; 
          item 'Select Active Rate File for Editing' dialog=d1; 
          item 'End this window' selection=endwdw; 
      item 'End this SAS session' selection=endsas; 
      selection endwdw 'end'; 
      selection endsas 'bye'; 

/**** the dialog d1 statement transfers control to the fileviewma program for MPC 
active rate file editing***/ 
/***** the actual rate selected is controlled by the radio button (rbutton) 
statements **********************/ 
/***** which depending on the button selected by the user substitutes the rate for 
the mafilename file*******/ 
/***** that is opened by the fileviewma 
program************************************************************/ 
 
 dialog d1 'end;pgm;include fileview;change mafilename "%1" all; 
               submit'; 
 
 text #1 @1 'Choose File :'; 
          radiobox default=1; 
             rbutton #2 @5 "%scan(&ratema1,-2,.)" ; 
             rbutton #2 @25 "%scan(&ratema2,-2,.)" ; 
             rbutton #3 @5 "%scan(&ratema3,-2,.)" ; 
             rbutton #3 @25 "%scan(&ratema4,-2,.)" ; 
             rbutton #4 @5 "%scan(&ratema5,-2,.)" ; 
             rbutton #4 @25 "%scan(&ratema6,-2,.)" ; 
             rbutton #5 @5 "%scan(&ratema7,-2,.)" ; 
             rbutton #5 @25 "%scan(&ratema8,-2,.)" ; 
             rbutton #6 @5 "%scan(&ratema9,-2,.)" ; 
             rbutton #6 @25 "%scan(&ratema10,-2,.)" ; 
             rbutton #7 @5 "%scan(&ratema11,-2,.)" ; 
             rbutton #7 @25 "%scan(&ratema12,-2,.)" ; 
             rbutton #8 @5 "%scan(&ratema13,-2,.)" ; 
             rbutton #8 @25 "%scan(&ratema14,-2,.)" ; 
             rbutton #9 @5 "%scan(&ratema15,-2,.)" ; 
             rbutton #9 @25 "%scan(&ratema16,-2,.)" ; 
             rbutton #10 @5 "%scan(&ratema17,-2,.)" ; 
             rbutton #10 @25 "%scan(&ratema18,-2,.)" ; 
             rbutton #11 @5 "%scan(&ratema19,-2,.)" ; 
             rbutton #11 @25 "%scan(&ratema20,-2,.)" ; 
    rbutton #12 @5 "%scan(&ratema21,-2,.)" ; 
             rbutton #12 @25 "%scan(&ratema22,-2,.)" ;  
    rbutton #13 @5 "%scan(&ratema23,-2,.)" ; 
             rbutton #13 @25 "%scan(&ratema24,-2,.)" ; 
    rbutton #14 @5 "%scan(&ratema25,-2,.)" ; 
             rbutton #14 @25 "%scan(&ratema26,-2,.)" ; 
    rbutton #15 @5 "%scan(&ratema27,-2,.)" ; 
             rbutton #15 @25 "%scan(&ratema28,-2,.)" ; 
             rbutton #16 @5 "%scan(&ratema29,-2,.)" ; 
             rbutton #16 @25 "%scan(&ratema30,-2,.)" ; 
    rbutton #17 @5 "%scan(&ratema31,-2,.)" ; 
             rbutton #17 @25 "%scan(&ratema32,-2,.)" ;  
    rbutton #18 @5 "%scan(&ratema33,-2,.)" ; 
             rbutton #18 @25 "%scan(&ratema34,-2,.)" ; 
    rbutton #19 @5 "%scan(&ratema35,-2,.)" ; 
             rbutton #19 @25 "%scan(&ratema36,-2,.)" ; 
    rbutton #20 @5 "%scan(&ratema37,-2,.)" ; 
             rbutton #20 @25 "%scan(&ratema38,-2,.)" ; 
             rbutton #21 @5 "%scan(&ratema39,-2,.)" ; 
             rbutton #21 @25 "%scan(&ratema40,-2,.)" ; 
 
 
 run; 
 quit; 
/**the ratefilelistma program updates the rate selection menu for rates found in the 
MPC active file directory*******/ 
/**the &ratema macro variables contains filenames generated from reading the MPC 
active directory which are used******/ 

/** to update the menu 
dynamically*************************************************************************
**********/ 
 
 %include filelist; 

 
The ratefilelistma program updates the rate selection menu for rates found in the MPC active file directory. This code 
is shown below. It creates the macro variables for the rate filenames found in the MPC Active file directory. The code 
uses the pipe command  and the DOS directory macro command set up in the initialization program above to get a list 
of  MPC active rate files. (%let filedirma=dir/b &maindir\rates\mpc\active\*rat.sas7bdat) 
 

/******************************************************************************/ 
/*  PROGRAM:  ratefilelistma                                                  */ 
/******************************************************************************/ 
/*  DATE CREATED:  October 2002                                               */ 
/*  BY:  Bland                                                                */ 
/******************************************************************************/ 
/*  DESCRIPTION:                                                              */ 
/*    THIS PROGRAM queries directories for rate file names                    */ 
/*    parses those names and stores them in global macro variables            */ 
/*    which are used to dynamically update the menus in the Rateman system    */ 
/******************************************************************************/ 
/*  INPUTS: Input File Directory in macro variable &filedirXX with XX defined */ 
/*  as follows:                                                               */ 
/*  aa- APC Active, ah- APC History, fa- Gulf Active, fh - Gulf History       */ 
/*  ga- GPC Active, gh- GPC History, ma- MPC Active, mh- MPC History          */ 
/*  sa- Savannah Active, sh-Savannah History                                  */ 
/*  oa- Other Company Active, oh- Other Company History                       */ 
/******************************************************************************/ 
/*  OUTPUTS: Global Macro Variables with Rate Filenames, &rateXX1 - &rateXX20 */ 
/*  with XX defined as above                                                  */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  EDITED BY:                             DATE:                              */ 
/******************************************************************************/ 
 
options nomlogic nomprint nomrecall nosymbolgen; 
*options mprint mlogic symbolgen; 
/*This code uses the windows pipe command to store the file directory information*/ 
/* from the directory stored in the input macro variable*/ 
 
filename ratelist pipe "&filedirma"; 
 
/*this code creates a temporary dataset,bill, that holds the filename information 
from the pipe*/ 
data bill; 
infile ratelist length=l; 
length filename $200; 
input  @; 
input @1 filename $varying200. l; 
filename=upcase(filename); 
run; 
 
filename ratelist clear; 
run; 
 
/*this code transposes the data in temp file bill and creates variables with the 
company file prefix stored in */ 
/* another temporary file bill2*/ 
proc transpose data=bill 
out=bill2 prefix=marat; 
var filename; 
run; 

/* this code uses the proc contents procedure to do a count of rate files and stores 
the numbers in variable varnum */ 
/* in a temporary file test*/ 
   proc contents noprint data=bill2 out=test(keep=varnum); 
   run; 
 
/* the output from test and bill2 are input into another temporary file bill3*/ 
/* the ceil function is used to determine the maximum of varnum and stored in the 
macro variable numrates*/ 
 
   data bill3; 
   set bill2 test; 
   nummax=ceil(varnum); 
      call symput('numrates',nummax); 
  
 run; 
 
 /* This macro code loops over the number of rates and creates individual file name 
global macro variables */ 
 /* that are passed to the menu programs, the maximum number of rates in a directory 
is 20 */ 
 %global ratema1 ratema2 ratema3 ratema4 ratema5 ratema6 ratema7 ratema8 ratema9 
ratema10 
  ratema11 ratema12 ratema13 ratema14 ratema15 ratema16 ratema17 ratema18 ratema19 
ratema20 
 ratema21 ratema22 ratema23 ratema24 ratema25 ratema26 ratema27 ratema28 ratema29 
ratema30 
 ratema31 ratema32 ratema33 ratema34 ratema35 ratema36 ratema37 ratema38 ratema39 
ratema40;  
  
   %macro bill; 
   
   %do k=1 %to %eval(&numrates); 
   %do i=%eval(&k-1) %to %eval(&k+1); 
    data bill3; 
   set bill2; 
   call symput("ratema&&i",trim(marat&&i)); 
    
 
     %let ratema&i=%str("&ratema&&i"); 
   
    %put _all_; 
      %end; 
   %end; 
 run; 
%mend bill; 
%put _all_; 
%bill; 
 
/* this code deletes the temporary data files*/ 
proc datasets; 
delete bill bill2 bill3 ; 
run; 

Once the user selects a rate using the radio button, it is loaded into PROC FSVIEW for viewing or editing by the user. 
The dialog statement changes the mafilename to the appropriate rate selection and FSVIEW presents the file to the 
user. When the file is closed, control returns to the main menu system. The fileviewma program accomplishes this 
task and the code is shown below. 
 

/******************************************************************************/ 
/*  PROGRAM:  fileviewma                                                      */ 
/******************************************************************************/ 
/*  DATE CREATED:  October 2002                                               */ 
/*  BY:  Bland                                                                */ 
/******************************************************************************/ 
/*  DESCRIPTION:                                                              */ 
/*    THIS PROGRAM is the file viewer program for MPC Active Rates            */ 
/*    Proc FSVIEW allows the user to view, edit, create and print rate files  */ 
/******************************************************************************/ 
/*  INPUTS: Rate File                                                         */ 
/*                                                                            */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  OUTPUTS: Edited Rate file or New Rate File                                */ 
/*           Printer Output of Rate if desired                                */ 
/*                                                                            */ 
/******************************************************************************/ 
/*  EDITED BY:                             DATE:                              */ 
/******************************************************************************/ 
libname mpcactiv "&vfiledirma"; 
/*    Proc FSVIEW allows the user to view, edit, create and print rate files  */ 
proc fsview data=mpcactiv.mafilename; 
run; 
/****This code returns control to the Rateman input screen**************/ 
filename driver "&maindir\rateman-driver.sas"; 
%include driver; 
run; 

 

USER TESTING OF CODE 

The Rate Manager system was developed over a 4-year period and is currently undergoing annual updates. We have 
developed the system one step at a time with user meetings to determine the requirements of the system before 
coding a section is begun. Coding is developed and then sent to the users to test. If changes are necessary, they are 
made and the code is sent back to the user for testing and this process repeats until the code is signed off on by the 
user team.  

 

USER ‘S GUIDE 
A comprehensive user’s guide was developed for the Rate Manager system. This 85 page document is a complete 
guide for the user and explains the details of each Rate Manager task and has sample input and expected output. 
The Guide was also user tested for readability and ease of use. The user’s guide is also available while running the 
system by selecting the Help menu, the user’s guide will pop up in a Microsoft Word document. 

RATE MANAGER OVERVIEW 
The Rate Manager system was developed to assist the Rate Analyst in the analysis of current rates and the design of 
new rates. It also allows for rate comparisons across companies and jurisdictions. Once a rate is entered into the 
system, the rate is then formulized. This process takes a given utility rate and creates a set of mathematical equations 
that describe the rate over discrete sections of the energy and demand space as represented by an X-Y graph where 
demand or kW is shown on the X axis and energy is shown on the Y axis. 
 
After a rate is formulized, it can be compared to another rate by the development of a differential chart. The 
differential chart is a set of equations that are developed by setting the rate formula for 2 rates equal to each other in 
each applicable section of the demand energy space and solving for what are called lines of equality. The line(s) of 
equality are lines of kW and kWh where the 2 rates are equal to each other. 
 
Rate Manager software produced the graph below and is an example of a differential chart and shows the line of 
equality (bold red line) between the 2 rates. The line of equality divides the demand-energy space into 2 regions 
where one rate is lower than the other. All graphs are created in HTML format for presentation on the WEB. 

The graph below shows the differential chart with customer demand and energy data overlay produced by Rate 
Manager software. The customers above the bold red line may need to be on rate X rather than rate Y. 

Rate Manager software also allows the user to plot the differential chart with contour lines showing the percentage 
cost differences between the 2 rates. This is shown below. 
 

The graph below is an example of the rate graphing capability of the Rate Manager software package. The software 
automatically determines (from rate inputs) regions in the Demand –Energy space where the price is constant. 
Vertical lines show demand breaks and horizontal lines show energy breaks and slanting lines hours use breaks.  

The Graph below shows an example of a rate profitability graph with customer data overlaid produced by Rate 
Manager software. Customers below the green contour line are profitable to serve. The contour graph can also be 
displayed as colored regions and the graph is also shown. 

Rate Manager software also allows the user to produce graphs at a given demand level showing bill comparisons for 
up to 6 rates compared at all hours use levels. 
 
  

The software also outputs tabular rate comparisons in HTML format for presentation on the WEB. A portion of the 
output is shown below. 
 

Bill Comparison for Rate X Versus Rate Y 

KW=100 

K
W 

Energ
y 

Hours 
Use 

 
RATE X  

SUMMER  
CUST 

TRANSFORMER 

 
RATE X  
WINTER  

CUST 
TRANSFORMER 

Bill 
Difference  

$ 

Percent 
Difference  

% 

100 5,000 50 $391.09 $365.54 ($25.55) (6.53) 

100 10,000 100 $764.56 $727.55 ($37.01) (4.84) 

100 15,000 150 $1,108.92 $1,049.01 ($59.91) (5.40) 

100 20,000 200 $1,453.27 $1,370.46 ($82.81) (5.70) 

 
: 

CONCLUSION  
This paper has shown that it is possible to develop a major user-friendly point and click application for non-SAS users 
in SAS®  using PROC Pmenu as the basis. This system has been used to design new electric utility rates and to 
determine the best rate for a utility customer. The system was developed with the user in mind and considerable user 
testing of the software at each stage of development was carried out. The software was presented at the 
Southeastern Electric Exchange's Rate section in 2005 and won first place for new innovative applications. 

